

# Voting power in the 15-EU

**J. M. Bilbao, J. R. Fernández, J. J. López**

```
In[1] := <<DiscreteMath`Combinatorica`
```

```
<<Graphics`Graphics`
```

```
<<Graphics`Graphics3D`
```

## Banzhaf index

The function `banzhafG` computes the generating function given a list of integer weights. The function `banzhafIndexPlus` computes the total Banzhaf power index of player  $i$  by summing the appropriate coefficients in this generating function. Dividing each player's index by the sum of all the indices gives the Banzhaf power distribution.

```
banzhafG[weights_List] := Times @@ (1+x^weights)
```

```
banzhafIndexPlus[i_, weights_List, q_] :=
Module[{delw, sw, g, coefi},
delw=Delete[weights, i];
sw=Apply[Plus, delw]+1;
g=banzhafG[delw];
coefi=CoefficientList[g, x];
Apply[Plus, coefi[[
Range[Max[1, q-weights[[i]]+1], Min[q, sw]]]]]
]
```

```
banzhafPowerPlus[weights_List, q_] :=
#/(Plus @@ #)& @ Table[banzhafIndexPlus[i, weights, q],
{i, Length[weights]}]
```

```
critical[weights_List, q] :=
Table[banzhafIndexPlus[i, weights, q], {i, Length[weights]}]
```

```
votesEU15={10, 10, 10, 10, 10, 8, 5, 5, 5, 5, 4, 4, 3, 3, 2};
```

```
votesEU15nice={29, 29, 29, 29, 27, 13, 12, 12, 12, 10, 10, 7, 7, 7, 4};
```

```
swingsold=critical[votesEU15, 62]
```

```
{1849, 1849, 1849, 1849, 1531, 973, 973, 973, 973, 793, 793, 595,
595, 595, 375}
```

```
swingsnice=critical[votesEU15nice,169]
```

```
{1989, 1989, 1989, 1989, 1853, 925, 869, 869, 869, 721, 721, 521,
 521, 521, 339}
```

```
banzhafPowerPlus[votesEU15,62]
```

```
{ $\frac{1849}{16565}$ ,  $\frac{1849}{16565}$ ,  $\frac{1849}{16565}$ ,  $\frac{1849}{16565}$ ,  $\frac{1531}{16565}$ ,  $\frac{973}{16565}$ ,  $\frac{973}{16565}$ ,  $\frac{973}{16565}$ ,  $\frac{973}{16565}$ ,
 $\frac{793}{16565}$ ,  $\frac{793}{16565}$ ,  $\frac{119}{3313}$ ,  $\frac{119}{3313}$ ,  $\frac{119}{3313}$ ,  $\frac{75}{3313}$ }
```

```
banEU15=SetPrecision[%,3]
```

```
{0.112, 0.112, 0.112, 0.112, 0.0924, 0.0587, 0.0587, 0.0587,
 0.0587, 0.0479, 0.0479, 0.0359, 0.0359, 0.0359, 0.0226}
```

```
banzhafPowerPlus[votesEU15nice,169]
```

```
{ $\frac{1989}{16685}$ ,  $\frac{1989}{16685}$ ,  $\frac{1989}{16685}$ ,  $\frac{1989}{16685}$ ,  $\frac{1853}{16685}$ ,  $\frac{185}{3337}$ ,  $\frac{869}{16685}$ ,  $\frac{869}{16685}$ ,  $\frac{869}{16685}$ ,
 $\frac{721}{16685}$ ,  $\frac{721}{16685}$ ,  $\frac{521}{16685}$ ,  $\frac{521}{16685}$ ,  $\frac{521}{16685}$ ,  $\frac{339}{16685}$ }
```

```
banEU15nice=SetPrecision[%,3]
```

```
{0.119, 0.119, 0.119, 0.119, 0.111, 0.0554, 0.0521, 0.0521, 0.0521,
 0.0432, 0.0432, 0.0312, 0.0312, 0.0312, 0.0203}
```

## Shapley-Shubik index

The number of coalitions of weight  $k$  and size  $j$  is the coefficient of  $x^k z^j$  in the generating function  $g(x,y)$  for the Shapley-Shubik index. The function `ssG` gives the polynomial  $g(x,y)$ . The function `ssPowerPlus` computes the Shapley-Shubik power distribution. We suppose that the simple game is superadditive, that is, winning disjoint coalitions are not possible.

```
ssG[weights_List]:=Times @@ (1+z x^weights)
```

```
ssPowerPlus[weights_List,q_Integer]:=
Module[{n=Length[weights],delw,sw,g,coefi,gg},
Table[delw=Delete[weights,i];
sw=Apply[Plus,delw]+1;
g=ssG[delw];
coefi=CoefficientList[g,x];
gg=Apply[Plus,coefi[[
Range[Max[1,q-weights[[i]]+1],Min[q,sw]]]]];
Sum[Coefficient[gg,z^j] j! (n-j-1)!,{j,n-1}],
{i,n}]/n!]
```

```
shaEU15=SetPrecision[ssPowerPlus[votesEU15,62],3]
```

```
{0.117, 0.117, 0.117, 0.117, 0.0955, 0.0552, 0.0552, 0.0552,  
 0.0552, 0.0454, 0.0454, 0.0353, 0.0353, 0.0353, 0.0207}
```

```
shaEU15nice=SetPrecision[ssPowerPlus[votesEU15nice,169],3]
```

```
{0.127, 0.127, 0.127, 0.127, 0.113, 0.0545, 0.0485, 0.0485, 0.0485,  
 0.0374, 0.0374, 0.0283, 0.0283, 0.0283, 0.0190}
```

## Weighting of votes and new power in the 15-EU on 2005

```
countries15EU={"Germany","United Kingdom","France","Italy","Spain",  
"Netherlands","Greece","Belgium","Portugal","Sweedden",  
"Austria","Denmark","Finland","Ireland","Luxembourg"};
```

```
TableForm[Transpose[{swingsold,swingsnice,swingsnice-swingsold}],
TableHeadings->{countries15EU,
{"Old Critical","Nice Critical","Difference"}}]
```

	Old Critical	Nice Critical	Difference
Germany	1849	1989	140
United Kingdom	1849	1989	140
France	1849	1989	140
Italy	1849	1989	140
Spain	1531	1853	322
Netherlands	973	925	-48
Greece	973	869	-104
Belgium	973	869	-104
Portugal	973	869	-104
Sweedden	793	721	-72
Austria	793	721	-72
Denmark	595	521	-74
Finland	595	521	-74
Ireland	595	521	-74
Luxembourg	375	339	-36

```
pop15EU={82.038,59.247,58.966,57.612,39.394,15.760,10.533,
10.213,9.980,8.854,8.082,5.313,5.160,3.744,0.429};
indexpop15EU=SetPrecision[(%/Plus @@ %),3]
```

```
{0.219, 0.158, 0.157, 0.153, 0.105, 0.0420, 0.0281, 0.0272, 0.0266,
0.0236, 0.0215, 0.0142, 0.0137, 0.00998, 0.00114}
```


```
TableForm[Transpose[{indexpop15EU,banEU15,banEU15nice,shaEU15,shaEU15nice}],
TableHeadings ->{countries15EU,
{"People","Old Ban","Nice Ban","Old Sha","Nice Sha"}}]
```

	People	Old Ban	Nice Ban	Old Sha	Nice Sha
Germany	0.219	0.112	0.119	0.117	0.127
United Kingdom	0.158	0.112	0.119	0.117	0.127
France	0.157	0.112	0.119	0.117	0.127
Italy	0.153	0.112	0.119	0.117	0.127
Spain	0.105	0.0924	0.111	0.0955	0.113
Netherlands	0.0420	0.0587	0.0554	0.0552	0.0545
Greece	0.0281	0.0587	0.0521	0.0552	0.0485
Belgium	0.0272	0.0587	0.0521	0.0552	0.0485
Portugal	0.0266	0.0587	0.0521	0.0552	0.0485
Sweedn	0.0236	0.0479	0.0432	0.0454	0.0374
Austria	0.0215	0.0479	0.0432	0.0454	0.0374
Denmark	0.0142	0.0359	0.0312	0.0353	0.0283
Finland	0.0137	0.0359	0.0312	0.0353	0.0283
Ireland	0.00998	0.0359	0.0312	0.0353	0.0283
Luxembourg	0.00114	0.0226	0.0203	0.0207	0.0190

```

EU15nice=BarChart3D[
  {Reverse[banEU15nice],Reverse[banEU15],Reverse[indexpop15EU],
  Reverse[shaEU15],Reverse[shaEU15nice]},
  Boxed -> True, BoxRatios -> {0.5,1,0.5},
  ViewPoint->{0.4,-2.2,0.5},Ticks ->
  {{{1,"Nice Banzhaf"},{2,"Old Banzhaf"},{3,"POPULATION"},
  {4,"Old Shapley"},{5,"Nice Shapley"}},
  {{1,"Luxembourg"},{2,"Ireland"},{3,"Finland"},
  {4,"Denmark"},{5,"Austria"},{6,"Sweedden"},
  {7,"Portugal"},{8,"Belgium"},{9,"Greece"},
  {10,"Netherlands"},{11,"Spain"},{12,"Italy"},
  {13,"France"},{14,"United Kingdom"},{15,"Germany"}},Automatic},
  DefaultFont->{"Times",8}];

```



```

propor15EU = Round[(pop15EU/Plus @@ pop15EU) 88]

```

```

{19, 14, 14, 14, 9, 4, 2, 2, 2, 2, 2, 1, 1, 1, 0}

```

```

Plus @@ %

```

```

87

```

```

propor15data=Transpose[{pop15EU,propor15EU}]

```

```


{{82.038, 19}, {59.247, 14}, {58.966, 14}, {57.612, 14},
 {39.394, 9}, {15.76, 4}, {10.533, 2}, {10.213, 2}, {9.98, 2},
 {8.854, 2}, {8.082, 2}, {5.313, 1}, {5.16, 1}, {3.744, 1},
 {0.429, 0}}

```

```
EU15data=Transpose[{pop15EU,votesEU15}]
```

```
{{82.038, 10}, {59.247, 10}, {58.966, 10}, {57.612, 10},
 {39.394, 8}, {15.76, 5}, {10.533, 5}, {10.213, 5}, {9.98, 5},
 {8.854, 4}, {8.082, 4}, {5.313, 3}, {5.16, 3}, {3.744, 3},
 {0.429, 2}}
```

```
DisplayTogether[ListPlot[EU15data,Frame->True,
FrameLabel->{"Population","Votes",
"POPULATION INDICES and 15-EUROPEAN UNION OLD VOTES",""},
FrameTicks->{{1,5,10,15,25,40,60,82},
{1,2,4,5,8,10,14,19}},
PlotStyle->{Hue[0.05],PointSize[0.009]}],
ListPlot[propor15data,PlotStyle->{Hue[0.6],PointSize[0.009]}],
ListPlot[propor15data,PlotJoined->True,PlotStyle->RGBColor[0,0,1]],
ListPlot[EU15data,PlotJoined->True,PlotStyle->RGBColor[1,0,0]]];
```



```
propor15niceEU = Round[(pop15EU/Plus @@ pop15EU) 237.8]
```

```
{52, 38, 37, 37, 25, 10, 7, 6, 6, 6, 5, 3, 3, 2, 0}
```

```
Plus @@ %
```

```
237
```

```
propor15nicedata=Transpose[{pop15EU,propor15niceEU}]
```

```
{{82.038, 52}, {59.247, 38}, {58.966, 37}, {57.612, 37},
 {39.394, 25}, {15.76, 10}, {10.533, 7}, {10.213, 6}, {9.98, 6},
 {8.854, 6}, {8.082, 5}, {5.313, 3}, {5.16, 3}, {3.744, 2},
 {0.429, 0}}
```


```
EU15nicedata=Transpose[{pop15EU,votesEU15nice}]
```

```
{{82.038, 29}, {59.247, 29}, {58.966, 29}, {57.612, 29},
 {39.394, 27}, {15.76, 13}, {10.533, 12}, {10.213, 12},
 {9.98, 12}, {8.854, 10}, {8.082, 10}, {5.313, 7}, {5.16, 7},
 {3.744, 7}, {0.429, 4}}
```

```

DisplayTogether[ListPlot[EU15nicedata, Frame->True,
FrameLabel->{"Population", "Votes",
"POPULATION INDICES and 15-EUROPEAN UNION NICE VOTES", ""},
FrameTicks->{{1, 5, 10, 15, 25, 40, 60, 82},
{2, 4, 6, 10, 12, 27, 29, 37, 52}},
PlotStyle->{Hue[0.05], PointSize[0.009]}],
ListPlot[propor15nicedata, PlotStyle->{Hue[0.6], PointSize[0.009]}],
ListPlot[propor15nicedata, PlotJoined->True, PlotStyle->RGBColor[0, 0, 1]],
ListPlot[EU15nicedata, PlotJoined->True, PlotStyle->RGBColor[1, 0, 0]];

```





## Banzhaf index for the meet of two games

```
banzhafTwoG[weights_List, pop_List] := Times @@ (1+x^weights y^pop)
```

```
banzhafTwoG[{4,6,8},{2,3,4}]
```

$$(1 + x^4 y^2) (1 + x^6 y^3) (1 + x^8 y^4)$$

```
banzhafTwoIndex[i_, weights_List, pop_List, q_, p_] :=
Module[{g, coefi, n, m, s1, s2},
g=banzhafTwoG[Delete[weights, i], Delete[pop, i]];
n=Exponent[g, x]+1; m=Exponent[g, y]+1;
coefi=CoefficientList[g, {x, y}]/.{}->Table[0, {m}];
s1=Apply[Plus, Flatten[coefi[[
Range[Max[1, q-weights[[i]]+1], n],
Range[Max[1, p-pop[[i]]+1], m]]]]];
s2=If[((q+1)>n) || ((p+1)>m), 0,
Apply[Plus, Flatten[coefi
[[Range[q+1, n], Range[p+1, m]]]]]];
s1-s2]
```

```
criticalTwo[weights_List, pop_List, q_, p_] :=
Table[banzhafTwoIndex[i, weights, pop, q, p], {i, Length[weights]}]
```

```
banzhafTwoPower[weights_List, pop_List, q_, p_] :=
#/(Plus @@ #)& @ Table[banzhafTwoIndex[i, weights, pop, q, p],
{i, Length[weights]}]
```

## Shapley-Shubik index for the meet of two games

```
ssTwoG[weights_List, pop_List] := Times @@ (1+x^weights y^pop z)
```

```
ssTwoPower[weights_List, pop_List, q_, p_] :=
Module[{n=Length[weights], g, dw, dp, s1, s2, gg, coefi},
Table[g=ssTwoG[Delete[weights, i], Delete[pop, i]];
dw=Exponent[g, x]+1; dp=Exponent[g, y]+1;
coefi=CoefficientList[g, {x, y}]/.{}->Table[0, {dp}];
s1=Apply[Plus, Flatten[coefi[[
Range[Max[1, q-weights[[i]]+1], dw],
Range[Max[1, p-pop[[i]]+1], dp]]]]];
s2=If[((q+1)>dw) || ((p+1)>dp), 0,
Apply[Plus, Flatten[coefi[[
Range[q+1, dw], Range[p+1, dp]]]]]];
gg=s1-s2;
Sum[Coefficient[gg, z, j] j! (n-j-1)!, {j, n-1}]/n!,
{i, n}]]
```

## Banzhaf and Shapley indices for the meet of three games

```

banzhafThreeG[weights_List, pop_List, members_List] :=
Times @@ (1 + x^weights*y^pop*z^members)

banzhaf3Index[i_,weights_List,pop_List,members_List,q_,p_,m_] :=
Module[{n = Length[weights],g,dw,dp,dm,s1,s2,gg,coefi,delwe,delpo,delm},
delwe = Delete[weights,i]; delpo = Delete[pop,i]; delm = Delete[members,i];
g = banzhafThreeG[delwe, delpo, delm];
dw = Plus @@ delwe + 1; dp = Plus @@ delpo + 1; dm = Plus @@ delm + 1;
coefi = CoefficientList[g, {x, y, z}] /. {} -> Table[0, {dp}, {dw}];
s1 = Plus @@ Flatten[coefi[[Range[Max[1, q - weights[[i]] + 1], dw],
Range[Max[1, p - pop[[i]] + 1], dp],
Range[Max[1, m - members[[i]] + 1], dm]]];
s2 = If[q + 1 > dw || p + 1 > dp || m + 1 > dm, 0,
Plus @@ Flatten[coefi[[Range[q + 1, dw],Range[p + 1, dp],
Range[m + 1, dm]]]]]; gg = s1 - s2]

banzhaf3swings[weights_List,pop_List,members_List,q_,p_,m_] :=
Table[banzhaf3Index[i,weights,pop,members,q,p,m],
{i, Length[weights]}]

banzhaf3Power[weights_List,pop_List,members_List,q_,p_,m_] :=
(#1/Plus @@ #1 & ) [Table[banzhaf3Index[i,weights,pop,members,q,p,m],
{i, Length[weights]}]]

Shapley3G[weights_List,pop_List,members_List] :=
Times @@ (1 + x^weights*y^pop*z^members*t)

Shapley3Index[weights_List,pop_List,members_List,q_,p_,m_] :=
Module[{n=Length[weights],g,dw,dp,dm,s1,s2,gg,coefi,delwe,delpo,delm},
Table[delwe=Delete[weights,i];delpo=Delete[pop,i];delm=Delete[members,i];
g=Shapley3G[delwe,delpo,delm];
dw = Plus @@ delwe+1; dp = Plus @@ delpo+1; dm = Plus @@ delm+1;
coefi = CoefficientList[g, {x, y, z}] /. {} -> Table[0, {dp}, {dw}];
s1 = Plus @@ Flatten[coefi[[Range[Max[1, q-weights[[i]]+1], dw],
Range[Max[1,p-pop[[i]]+1], dp],Range[Max[1,m-members[[i]]+1], dm]]];
s2 = If[q + 1 > dw || p + 1 > dp || m + 1 > dm, 0,
Plus @@ Flatten[coefi[[Range[q+1,dw],Range[p+1,dp],Range[m+1,dm]]]]];
gg = s1 - s2; Sum[Coefficient[gg,t,j] j! (n-j-1)!,{j,0,n-1}]/n!,{i,n}]]

```

## Nice swings in the 15-EU on 2005

The total number of coalitions in the 15 European Union is 32768. We calculate the number of winning coalitions to which the country  $i$  is critical, that is, his defection implies that the coalition to become losing.

The total number of winning coalitions to which a country is critical, under the votes Nice rule, is  $T = 15848$ .

The total number of winning coalitions to which a country is critical, under the majority Nice rule, is  $T = 15767$ .

The total number of winning coalitions to which a country is critical, under the two thirds Nice rule, is  $T = 13243$ .

```

members15=Table[1,{15}];

swings1=critical[votesEU15nice,169]
{1989, 1989, 1989, 1989, 1853, 925, 869, 869, 869, 721, 721, 521,
 521, 521, 339}

Plus @@ %
16685

swingsTwo1=criticalTwo[votesEU15nice,members15,169,8]
{1989, 1989, 1989, 1989, 1853, 925, 869, 869, 869, 721, 721, 521,
 521, 521, 339}

swingsTwo2=criticalTwo[votesEU15nice,members15,169,10]
{1637, 1637, 1637, 1637, 1551, 943, 887, 887, 887, 791, 791, 653,
 653, 653, 513}

Plus @@ %
15757

weight15EU=Round[(pop15EU/Plus @@ pop15EU) 999.6]
{218, 158, 157, 153, 105, 42, 28, 27, 27, 24, 22, 14, 14, 10, 1}

Plus @@ %
1000

swingsThree1=banzhaf3swings[votesEU15nice,weight15EU,members15,169,620,8]
{2004, 1984, 1984, 1984, 1838, 910, 854, 854, 854, 712, 712, 512,
 512, 512, 324}

Plus @@ %
16550

swingsThree2=banzhaf3swings[votesEU15nice,weight15EU,members15,169,620,10]
{1652, 1632, 1632, 1632, 1536, 928, 872, 872, 872, 782, 782, 644,
 644, 644, 498}

Plus @@ %
15622

```

```
TableForm[Transpose[{swings1, swingsTwo1, swingsThree1, swingsThree1-swings1}],
TableHeadings->{countries15EU,
{"Swings 1", "Swings 2", "Swings 3", "Sw3-Sw1"}}]
```

	Swings 1 1989	Swings 2 1989	Swings 3 2004	Sw3-Sw1 15
Germany				
United Kingdom	1989	1989	1984	-5
France	1989	1989	1984	-5
Italy	1989	1989	1984	-5
Spain	1853	1853	1838	-15
Netherlands	925	925	910	-15
Greece	869	869	854	-15
Belgium	869	869	854	-15
Portugal	869	869	854	-15
Sweedon	721	721	712	-9
Austria	721	721	712	-9
Denmark	521	521	512	-9
Finland	521	521	512	-9
Ireland	521	521	512	-9
Luxembourg	339	339	324	-15

```
TableForm[Transpose[{swings1, swingsTwo2, swingsThree2,
swingsThree2-swings1, swingsThree2-swingsTwo2}],
TableHeadings->{countries15EU,
{"Swings1", "Swings2/3", "Swings3", "Sw3-Sw1", "Sw3-Sw2/3"}}]
```

	Swings1	Swings2/3	Swings3	Sw3-Sw1	Sw3-Sw2/3
Germany	1989	1637	1652	-337	15
United Kingdom	1989	1637	1632	-357	-5
France	1989	1637	1632	-357	-5
Italy	1989	1637	1632	-357	-5
Spain	1853	1551	1536	-317	-15
Netherlands	925	943	928	3	-15
Greece	869	887	872	3	-15
Belgium	869	887	872	3	-15
Portugal	869	887	872	3	-15
Sweedden	721	791	782	61	-9
Austria	721	791	782	61	-9
Denmark	521	653	644	123	-9
Finland	521	653	644	123	-9
Ireland	521	653	644	123	-9
Luxembourg	339	513	498	159	-15

## Power with the votes, majority and people Nice rules

```
ban1EU15=SetPrecision[banzhafPowerPlus[votesEU15nice,169],3]
```

```
{0.119, 0.119, 0.119, 0.119, 0.111, 0.0554, 0.0521, 0.0521, 0.0521,
0.0432, 0.0432, 0.0312, 0.0312, 0.0312, 0.0203}
```

```
ban2aEU15=SetPrecision[banzhafTwoPower[votesEU15nice,members15,169,8],3]
```

```
{0.119, 0.119, 0.119, 0.119, 0.111, 0.0554, 0.0521, 0.0521, 0.0521,
0.0432, 0.0432, 0.0312, 0.0312, 0.0312, 0.0203}
```

```
ban3aEU15=SetPrecision[banzhaf3Power[
votesEU15nice,weigth15EU,members15,169,620,8],3]
```

```
{0.121, 0.120, 0.120, 0.120, 0.111, 0.0550, 0.0516, 0.0516, 0.0516,
0.0430, 0.0430, 0.0309, 0.0309, 0.0309, 0.0196}
```


```
TableForm[Transpose[{{indexpop15EU,ban1EU15,ban2aEU15,ban3aEU15}},  
TableHeadings->{countries15EU,{"People","Banz Votes",  
"Banz V+M","Banz V+M+P"}}]
```

	People	Banz Votes	Banz V+M	Banz V+M+P
Germany	0.219	0.119	0.119	0.121
United Kingdom	0.158	0.119	0.119	0.120
France	0.157	0.119	0.119	0.120
Italy	0.153	0.119	0.119	0.120
Spain	0.105	0.111	0.111	0.111
Netherlands	0.0420	0.0554	0.0554	0.0550
Greece	0.0281	0.0521	0.0521	0.0516
Belgium	0.0272	0.0521	0.0521	0.0516
Portugal	0.0266	0.0521	0.0521	0.0516
Sweedn	0.0236	0.0432	0.0432	0.0430
Austria	0.0215	0.0432	0.0432	0.0430
Denmark	0.0142	0.0312	0.0312	0.0309
Finland	0.0137	0.0312	0.0312	0.0309
Ireland	0.00998	0.0312	0.0312	0.0309
Luxembourg	0.00114	0.0203	0.0203	0.0196

```

EU15nice3a=BarChart3D[{Reverse[indexpop15EU], Reverse[ban1EU15],
Reverse[ban2aEU15], Reverse[ban3aEU15]},
Boxed -> True, BoxRatios -> {0.5, 1, 0.5},
ViewPoint->{0.4, -2.2, 0.5}, Ticks ->{{1, "POPULATION"},
{2, "BANZHAF (V)"}, {3, "BANZHAF (V+M)"}, {4, "BANZHAF (V+M+P)"},
{{1, "Luxembourg"}, {2, "Ireland"}, {3, "Finland"},
{4, "Denmark"}, {5, "Austria"}, {6, "Sweedn"},
{7, "Portugal"}, {8, "Belgium"}, {9, "Greece"},
{10, "Netherlands"}, {11, "Spain"}, {12, "Italy"},
{13, "France"}, {14, "United Kingdom"}, {15, "Germany"}}, Automatic},
DefaultFont->{"Times", 8}];

```



## Power with the votes, two thirds and people Nice rules

```
ban2bEU15=SetPrecision[banzhafTwoPower[votesEU15nice, members15, 169, 10], 3]
```

```
{0.104, 0.104, 0.104, 0.104, 0.0984, 0.0598, 0.0563, 0.0563,
0.0563, 0.0502, 0.0502, 0.0414, 0.0414, 0.0414, 0.0326}
```

```
ban3bEU15=SetPrecision[banzhaf3Power[
votesEU15nice, weight15EU, members15, 169, 620, 10], 3]
```

```
{0.106, 0.104, 0.104, 0.104, 0.0983, 0.0594, 0.0558, 0.0558,
0.0558, 0.0501, 0.0501, 0.0412, 0.0412, 0.0412, 0.0319}
```

```
TableForm[Transpose[{{indexpop15EU,ban1EU15,ban2bEU15,ban3bEU15}},
TableHeadings->{countries15EU,{"People","Banz Vote",
"Banz V+(2/3)","Banz V+(2/3)+P"}}]
```

	People	Banz Vote	Banz V+(2/3)	Banz V+(2/3)+P
Germany	0.219	0.119	0.104	0.106
United Kingdom	0.158	0.119	0.104	0.104
France	0.157	0.119	0.104	0.104
Italy	0.153	0.119	0.104	0.104
Spain	0.105	0.111	0.0984	0.0983
Netherlands	0.0420	0.0554	0.0598	0.0594
Greece	0.0281	0.0521	0.0563	0.0558
Belgium	0.0272	0.0521	0.0563	0.0558
Portugal	0.0266	0.0521	0.0563	0.0558
Sweedn	0.0236	0.0432	0.0502	0.0501
Austria	0.0215	0.0432	0.0502	0.0501
Denmark	0.0142	0.0312	0.0414	0.0412
Finland	0.0137	0.0312	0.0414	0.0412
Ireland	0.00998	0.0312	0.0414	0.0412
Luxembourg	0.00114	0.0203	0.0326	0.0319



```

EU15nice3a=BarChart3D[{Reverse[indexpop15EU], Reverse[ban1EU15],
Reverse[ban2bEU15], Reverse[ban3bEU15]},
Boxed -> True, BoxRatios -> {0.5, 1, 0.5},
ViewPoint->{0.4, -2.2, 0.5}, Ticks ->{{1, "POPULATION"},
{2, "BANZHAF (V)"}, {3, "BANZHAF (V+(2/3))"}, {4, "BANZHAF (V+(2/3)+P)"},
{{1, "Luxembourg"}, {2, "Ireland"}, {3, "Finland"},
{4, "Denmark"}, {5, "Austria"}, {6, "Sweden"},
{7, "Portugal"}, {8, "Belgium"}, {9, "Greece"},
{10, "Netherlands"}, {11, "Spain"}, {12, "Italy"},
{13, "France"}, {14, "United Kingdom"}, {15, "Germany"}}, Automatic},
DefaultFont->{"Times", 8}];
    
```

